

# AUTOMATION AND VISUALIZATION OF DISTRACTOR ANALYSIS USING SAS/GRAPH

Chong Ho Yu, Ph.D., Arizona State University, Tempe, AZ

## ABSTRACT

In multiple-choice items of a test, it is desirable to insert plausible distractors (wrong answers) in order to increase the item difficulty level. According to the Item Response Theory, students who master the subject matter should have a higher probability to select the right answer and a lower probability to choose the wrong options. In contrast, examinees who do not master the skill should show the opposite pattern. This paper discusses how SAS programming is used to automate and visualize distractor analysis based upon empirical data and modeling derived from the Item Response Theory.


## INTRODUCTION

In multiple-choice items, it is desirable to insert plausible distractors (wrong answers) in order to increase the item difficulty level. According to the Item Response Theory (IRT), student's proficiency is estimated in terms of theta. Students who master the subject matter belong to the high theta group (above zero). Those students should have a higher probability of selecting the right answer, and have a lower probability of choosing the wrong options. In contrast, examinees who belong to the low theta group (below zero) should show the opposite pattern. This type of analysis is named option analysis, also known as distractor analysis.

There are two approaches to conduct option analysis, namely, the empirical approach and the model-based approach. This article will discuss both approaches using SAS.\*

## EMPIRICAL APPROACH: RELATIVE FREQUENCY AND THETA

Figure 1. Line plot of probability by theta.



The empirical approach uses the information of the relative frequency of choosing various options and the theta as indicated in the following SAS plot (Figure 1). This can help the instructor

to visualize whether distractors and the right answer behave in the way the item writer expected.

In this example (see Figure 1), the right answer is Option 3 (represented by the green symbol and the green line). Students who do not possess the knowledge (theta Below zero) required for answering this question fail to select Option 3, and students who master the domain knowledge (theta above zero) tend to choose the correct answer. Nonetheless, this item seems to be an easy question, because even an average student (theta = 0) has .91 probability to answer the question correctly.

It is important to note that this probability is conditional on theta. To be specific, it is derived from the ratio between the number of subjects in the same theta level who chose a particular option and the number of all subjects who belong to the same theta level.

For example, Table 1 shows that there are 250 students in the group of theta = 0. They are considered average students if the mean of the student ability is set to zero in the IRT program. Among these 250 students, 227 of them selected Option 3. Thus, the conditional probability is the column percentage = 227/255 = .91. (SAS converts the probability to percentage .91 = 90.80%.)


Table 1. Frequency table of conditional probability.

Frequency Col Pct	Table of q1 by t2r				
	t2r				Total
q1	-2	-1	0	1	
1	3 21.43	20 17.70	9 3.60	2 1.45	34
2	3 21.43	10 8.85	8 3.20	0 0.00	21
3	5 35.71	71 62.83	227 90.80	136 98.55	439
4	3 21.43	12 10.62	6 2.40	0 0.00	21
<b>Total</b>	14	113	250	138	515

In this context, the conditional probability is analogous to the column percentage (col pct).  $\text{Prob}(\text{option}=1|\text{theta}=-3) + \text{Prob}(\text{option}=2|\text{theta}=-3) + \text{Prob}(\text{option}=3|\text{theta}=-3) + \text{Prob}(\text{option}=4|\text{theta}=-3)$  should be equal to one. This can be verified by summing 3.60%, 3.20%, 90.80%, and 2.40% in the third column (the sum must be 100%). The same principle is applied to the conditional probability of other theta levels.


If the SAS device option is "ActiveX" or "Java," the plot is interactive. In other words, you can transform the chart type by right-clicking on the graph (see Figure 2).

Figure 2. Changing the chart type



If the line plot is transformed into a stacked bar graph, all bars should have equal height and occupy the entire vertical axis, as shown in Figure 3. This can help you to verify the accuracy of the distractor plot.

Figure 3. Bar graph of sum of probabilities.



The following source code is used to visualize the frequency table in SAS/Graph:

```
/* output the graph in a webpage using
Output Delivery System */
ods html
file="distractor_empirical.html";

/* set the graphical options */
options device=activex;
axis1 order=(0 to 1 by .1);
symbol1 i=join c=red width=2;
symbol2 i=join c=blue width=2;
symbol3 i=join c=green width=2;
symbol4 i=join c=black width=2;

/* Round off the theta value to integers
*/
data new; set theta;
theta = round(t2);
title "Distractor analysis";
```

```
/* Output the frequency count */
proc freq data=new; tables item1 * theta
/nopercent norow out=t;
proc sort; by theta;

/* Sum the frequency count by theta */
proc summary data=t; var count;
class theta;
output out=tb sum=sum;

/* Merge the proc summary and proc freq
output. compute the conditional prob */
data tmp; merge t tb; by theta;
if _type_ = 0 then delete;
prob = (count/sum);

/* Plot the data */
proc gplot; plot prob*theta = item1
/ vaxis=axis1;
run;
ods html close;
quit;
```

The procedures are explained in the following:


1. Use a software program for Item Response Theory (e.g., Bilog, Bilog MG, Winsteps) to estimate the theta for each subject. Some programs that are capable of constructing one-, two-, and three-parameter models would return three different thetas, but some would produce only one theta from the one-parameter model.
2. Merge the theta and the item responses into one file. Round off the value of the theta to integers.
3. In SAS use PROC FREQ to output the frequency table. Please notice that the output from PROC FREQ includes the data of frequency and total percentage, but not row percentage or column percentage.
4. Use PROC SUMMARY to obtain the number of observations by theta.
5. Merge the output from PROC FREQ and the output from PROC SUMMARY. Then compute the conditional probability.
6. Use SAS/Graph to plot the distractor analysis and output the graph as an ActiveX object, a Java object, or a GIF image.

## MODEL-BASED APPROACH: RESPONSE FUNCTION

The model-based approach presents option analysis by showing response functions of all options in a multiple-choice item (Thissen & Steinberg, 1996). The response functions are represented by trace lines in the plot (see Figure 4). It is important to note that the response function has probabilistic interpretations, and thus it does not describe an observable relative frequency probability. As you notice, although Figure 2 and Figure 4 are based on the same data set, the appearance of the model-based graphic is very different from that of its empirical counterpart. For instance, the theta values of in the empirical graph range from  $-2$  to  $+1$ , while the theta values in the model-based plot extend to  $+5$  and  $-5$ . Actually no students

have such extreme theta values. The model indicates what the probabilities would have been if there were very good and very poor students, respectively.

Figure 4. Model-based visualization of option analysis



Implementation of model-based option analysis was introduced by Dr. Cody Ding (2001). In this approach, each option is treated as the “right answer” and the item parameters for each option are estimated by the IRT program. Then Item Characteristic Curves (ICC), or trace lines, are plotted in SAS using this information. The following source code can generate the visual representation of the distractor model in SAS/Graph:

```

/* combine the four output files of
estimated parameters */
data combine; set one two three four;
/* convert the logistic metric to the
normal metric, if necessary */
d = 1.7;
/* create a variable called theta and
populate it with the range from +5 to -5
*/
do theta = -5 to 5 by 1; output;
end;

/* compute the probabilities under each
theta level, reverse the p value so that
the direction of the trace line of wrong
answers are reversed */
data new; set combine;
p = 1 - ((exp(d*a*(theta - b))) / (1 +
exp(d*a*(theta - b))));

data new2; set new;
/* put the p value and the trace line of
the right answer into the right
direction */
if item=1 and option=3 then p = 1 -p;

/* plot the trace line */
ods html file="distractor_model.html";
options device=activex;
axis1 order=(0 to 1 by .1);
axis2 order=(-5 to 5 by 1);
proc gplot; by item;
plot p*theta = option / vaxis=axis1

```

```

haxis=axis2;
symbol1 i=join v=a c=red width=2;
symbol2 i=join v=b c=blue width=2;
symbol3 i=join v=c c=green width=2;
symbol4 i=join v=d c=black width=2;
run;
ods html close;
quit;

```


The procedures are explained in the following:

1. Change the key to make each option as the right answer. Run the IRT program n times (n is the number of options in the item) using different keys.
2. Combine the item parameters in SAS. In this example, two parameters (a and b) are used. These parameters determine the curve height in each theta level.
3. Compute the probability of choosing the option in each theta level. Reverse the curve for the wrong answers.
4. Use SAS/Graph to plot the distractor analysis and output the graph as an ActiveX object, a Java object, or a GIF image.

## CONCLUSION

It is difficult to conduct option analysis using numeric results only, such as viewing a frequency table (e.g. Table 1). Visualization of option analysis can help item writers to examine whether different options behave in the way they expected and to revise the problematic options if necessary. Although it seems to be a tedious task to generate a line plot for every item, especially when a test is composed of many questions, both empirical and model-based approaches can be easily automated by scripting languages such as Perl and SAS Macro (Yu, 2001; Yu, Bhatia, Cody, & Wadhvani, 2002).

Figure 5. Option analysis in Winsteps



## NOTE

- \* Winsteps (Rasch measurement software and publications, 2002) could perform option analysis by showing the frequency of each option conditional upon the measure (theta), as shown in Figure 5. But the graph is ASCII-based and each option is shown in a separate plot. Nonetheless, it is a built-in feature of Winsteps, which requires no programming.

## REFERENCES

- Ding, C. (2001). Personal communication.  
Rasch measurement software and publications (2002). Winsteps.  
[On-line] Available: <http://www.winsteps.com>
- Thissen, D., & Steinberg, L. (1996). A response model for multiple-choice items. In W. J. van der Linden & R. K. Hambleton (Eds.), Handbook of modern item response theory (pp. 51-65). New York: Springer.
- Yu, C. H. (2001 April). Developing data systems to support the analysis and development of large-scale, on-line assessment. Paper presented at the Annual Meeting of the American Educational Researcher Association, Seattle, WA.
- Yu, C. H., Bhatia, S., Cody, D., & Wadhvani, V. (2002 March). Automation and visualization of test item analysis. Paper presented at the 21st Annual Microcomputer in Education Conference, Tempe, AZ.

## CONTACT INFORMATION

Chong Ho Yu, Ph.D., CCNA, MCSE, CNE  
Psychometrician  
PO Box 612  
Tempe AZ 85280  
USA  
602-778-2722  
Email: [asumain@yahoo.com.hk](mailto:asumain@yahoo.com.hk)  
Website: <http://seamonkey.ed.asu.edu/~alex>